TM Modes of a Cylindrical Waveguide – Dr Colton, Winter 2018

In[44]:=
```
uαn[α_, n_] = BesselJZero[α, n];

(* for dimension of R = 10 cm *)
R = 0.10;
c = 3*^8;
wcutoff[alpha_, n_] := uαn[alpha, n] c / R
cutofftable = Table[wcutoff[alpha, n], {alpha, 0, 3}, {n, 1, 4}] ;
cutofftable // MatrixForm
cutofftable  // Flatten // Sort
```

I'm using a size of $R =$ 10 cm (chosen arbitrarily).

These are the cutoff frequencies of the first 16 modes

first in table form…

Out[49]//MatrixForm=

$$\begin{pmatrix} 7.21448\times10^{9} & 1.65602\times10^{10} & 2.59612\times10^{10} & 3.53746\times10^{10} \\ 1.14951\times10^{10} & 2.10468\times10^{10} & 3.05204\times10^{10} & 3.99711\times10^{10} \\ 1.54069\times10^{10} & 2.52517\times10^{10} & 3.48595\times10^{10} & 4.43879\times10^{10} \\ 1.91405\times10^{10} & 2.92831\times10^{10} & 3.90456\times10^{10} & 4.86704\times10^{10} \end{pmatrix}$$

…and then in list form

Out[50]= $\{7.21448\times10^{9}, 1.14951\times10^{10}, 1.54069\times10^{10}, 1.65602\times10^{10}, 1.91405\times10^{10}, 2.10468\times10^{10}, 2.52517\times10^{10}, 2.59612\times10^{10}, 2.92831\times10^{10}, 3.05204\times10^{10}, 3.48595\times10^{10}, 3.53746\times10^{10}, 3.90456\times10^{10}, 3.99711\times10^{10}, 4.43879\times10^{10}, 4.86704\times10^{10}\}$

These are the $k(\omega)$ dispersion relations for the first 16 modes.

In[54]:=
```
k[w_, alpha_, n_] := Sqrt[w^2 / c^2 - wcutoff[alpha, n]^2 / c^2]
Table[k[w, alpha, n], {alpha, 0, 3}, {n, 1, 4}] // Flatten // Sort // Reverse
Plot[%, {w, 0, 5*^10}, ImageSize → Large]
```

Out[55]=

$$\left\{\sqrt{-578.319+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-1468.2+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-2637.46+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-3047.13+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-4070.65+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-4921.85+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-7085.+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-7488.7+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-9527.76+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-10\,349.9+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-13\,502.1+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-13\,904.+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-16\,939.5+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-17\,752.1+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-21\,892.+\frac{w^2}{90\,000\,000\,000\,000\,000}}, \sqrt{-26\,320.1+\frac{w^2}{90\,000\,000\,000\,000\,000}}\right\}$$

(Plot is on next page.)



Note that these are the FIRST 16 modes, in the sense that α goes from 0 to 3 and n goes from 1 to 4, but they are not necessarily the LOWEST 16 modes. For example, the $(\alpha = 4, n = 1)$ mode is lower than many of these that are shown (with its $\omega_{cutoff} = 2.28 \times 10^{10}$ rad/s).

Pictures of E_z for the first 16 modes

In[63]:=
```
f[x_, y_, alpha_, n_] := BesselJ[alpha, Sqrt[(x^2 + y^2)] uαn[alpha, n] / R] Cos[alpha Arg[x + I y]] // N
Table[Plot3D[Evaluate[f[x, y, alpha, n]], {x, -0.1, 0.1}, {y, -0.1, 0.1},
   RegionFunction → Function[{x, y, z}, x^2 + y^2 < 0.1^2], PlotRange → All] , {alpha, 0, 3}, {n, 1, 4}]  //
 TableForm
```

Out[64]//TableForm=